

Nevtralen vodnik pravokotnega preseka v homogenem elektrostaticnem polju

Elektrostaticno polje. Koncept enojnega sloja.

Pisava, ki bo uporabljena na slikah:

```
Times14 = BaseStyle → {FontFamily → "Times", FontSize → 14};
```

■ Vhodni podatki

Dielektricnost praznega prostora:

$$\epsilon_0 = 8.854 \times 10^{-12};$$

Koordinatni sistem izberimo tako, da je os z vzporedna z vodnikom ter da gre skozi njegov spodnji levi vogal. Os x naj kaže v desno, os y pa gor.

Sirina in visina preseka vodnika:

$$a = 0.01;$$

$$b = 0.02;$$

Elektricna poljska jakost prvotnega homogenega polja, ki je usmerjeno v smeri osi x :

$$E_0 = 100;$$

■ Segmentiranje oboda preseka vodnika

Število segmentov na horizontalnih stranicah oboda vodnika in dolzine teh segmentov:

$$n_x = 2;$$

$$\delta l_x = a / n_x$$

$$0.005$$

Število segmentov na vertikalnih stranicah oboda vodnika in dolzine teh segmentov:

$$n_y = 2;$$

$$\delta l_y = b / n_y$$

$$0.01$$

Namenoma smo uporabili premajhno število segmentov ($n_x = 2$ in $n_y = 2$), da bo lažje slediti izračunom. Pozneje (pri resnem izračunu polja) bomo to število povečali.

Celotno število segmentov na obodu vodnika:

$$N_0 = 2 * (n_x + n_y)$$

$$8$$

Indeksiranje segmentov: segmente začnimo šteti v spodnjem levem vogalu oboda vodnika in nadaljujmo v matematični pozitivni smeri (nasprotno od smeri vrtenja urinega kazalca).

Mnozica (indeksov) segmentov na spodnji stranici oboda vodnika:

$$M_1 = \text{Table}[j, \{j, n_x\}]$$

$$\{1, 2\}$$

Mnozica (indeksov) segmentov na desni stranici oboda vodnika:

```
M2 = Table[j, {j, nx + 1, nx + ny}]
{3, 4}
```

Mnozica (indeksov) segmentov na zgornji stranici oboda vodnika:

```
M3 = Table[j, {j, nx + ny + 1, 2 * nx + ny}]
{5, 6}
```

Mnozica (indeksov) segmentov na levi stranici oboda vodnika:

```
M4 = Table[j, {j, 2 * nx + ny + 1, N0}]
{7, 8}
```

Matrika (oz. vektor) dolzin segmentov:

```
 $\delta l = \text{Table}[\text{Which}[\text{MemberQ}[\text{M1} \cup \text{M3}, j], \delta l_x, \text{MemberQ}[\text{M2} \cup \text{M4}, j], \delta l_y], \{j, \text{N0}\}]$ 
{0.005, 0.005, 0.01, 0.01, 0.005, 0.005, 0.01, 0.01}
```

Matriki (oz. vektorja) koordinat srediscnih tock segmentov:

```
 $\mathbf{xC} = \text{Table}[\text{Which}[\text{MemberQ}[\text{M1}, j], (j - 1 / 2) * \delta l_x, \text{MemberQ}[\text{M2}, j], a, \text{MemberQ}[\text{M3}, j], (2 * nx + ny - j + 1 / 2) * \delta l_x, \text{MemberQ}[\text{M4}, j], 0], \{j, \text{N0}\}]$ 
{0.0025, 0.0075, 0.01, 0.01, 0.0075, 0.0025, 0, 0}

 $\mathbf{yC} = \text{Table}[\text{Which}[\text{MemberQ}[\text{M1}, j], 0, \text{MemberQ}[\text{M2}, j], (j - nx - 1 / 2) * \delta l_y, \text{MemberQ}[\text{M3}, j], b, \text{MemberQ}[\text{M4}, j], (N0 - j + 1 / 2) * \delta l_y], \{j, \text{N0}\}]$ 
{0, 0, 0.005, 0.015, 0.02, 0.02, 0.015, 0.005}
```

Matrika (oz. vektor) enotskih vektorjev normale na površino vodnika v srediscnih tockah segmentov (eksterna normala: kaze iz območja vodnika v zunanost):

```
 $\mathbf{nC} = \text{Table}[\text{Which}[\text{MemberQ}[\text{M1}, j], \{0, -1\}, \text{MemberQ}[\text{M2}, j], \{1, 0\}, \text{MemberQ}[\text{M3}, j], \{0, 1\}, \text{MemberQ}[\text{M4}, j], \{-1, 0\}], \{j, \text{N0}\}]$ 
{{0, -1}, {0, -1}, {1, 0}, {1, 0}, {0, 1}, {0, 1}, {-1, 0}, {-1, 0}}
```

Normala je vektor; prva komponenta je x komponenta, druga pa y .

Distancni vektor med srediscnima tockama i -tega in j -tega segmenta ter njegova dolzina (absolutna vrednost):

```
PC[i_, j_] := {xC[[j]] - xC[[i]], yC[[j]] - yC[[i]]}
absPC[i_, j_] :=  $\sqrt{(\mathbf{xC}[[j]] - \mathbf{xC}[[i]])^2 + (\mathbf{yC}[[j]] - \mathbf{yC}[[i]])^2}$ 
```

■ Normalna komponenta prvotnega homogenega polja na površini vodnika

Matrika (oz. vektor) normalnih komponent prvotnega polja v srediscnih tockah segmentov:

```
 $\mathbf{E0n} = \text{Table}[\text{Which}[\text{MemberQ}[\text{M1}, j], 0, \text{MemberQ}[\text{M2}, j], \mathbf{E0}, \text{MemberQ}[\text{M3}, j], 0, \text{MemberQ}[\text{M4}, j], -\mathbf{E0}], \{j, \text{N0}\}]$ 
{0, 0, 100, 100, 0, 0, -100, -100}
```

■ Sistem enacb za izracun porazdelitve elektrine na površini vodnika in resitev tega sistema

Izracun smo opravili z naslednjim številom segmentov na horizontalnih oz. vertikalnih stranicah oboda vodnika:

```

nx = 100;
ny = 200;

```

Matrika koeficientov sistema enacb:

```

A = Table[If[i == j, 1, -(nC[[j]].PC[i, j]) *  $\frac{\delta l[[i]]}{\pi * (\text{absPC}[i, j])^2}$ ], {j, N0}, {i, N0}];

```

Matrika (oz. vektor) znanih vrednosti (desnih strani) sistema enacb:

```

B = Table[2 * ε0 * E0n[[j]], {j, N0}];

```

Resitev sistema - matrika (oz. vektor) ploskovnih gostot elektrine na segmentih:

```

σ = LinearSolve[A, B];

```


■ Risanje porazdelitve ploskovne gostote elektrine na površini vodnika

Porazdelitev elektrine na spodnji stranici oboda vodnika:

```

ListPlot[Table[{100 xC[[i]], σ[[i]] * 109}, {i, nx}], PlotRange → All, Frame → True,
  Axes → None, Times14, FrameLabel → {"x / cm", "σ / (nC/m2)", "", ""}]

```




Porazdelitev elektrine na zgornji stranici oboda vodnika:

```


ListPlot[Table[{100 xC[[i]], σ[[i]] * 109}, {i, nx + ny + 1, 2 nx + ny}], PlotRange → All,
  Frame → True, Axes → None, Times14, FrameLabel → {"x / cm", "σ / (nC/m2)", "", ""}]

```




Porazdelitev elektrine na desni stranici oboda vodnika:

```
ListPlot [Table[{100 yC[[i]],  $\sigma[[i]] * 10^9$ }, {i, nx + 1, nx + ny}], PlotRange -> {0, 3.7},
Frame -> True, Axes -> None, Times14, FrameLabel -> {"y / cm", " $\sigma / (nC/m^2)$ ", "", ""}]
```



Porazdelitev elektrine na levi stranici oboda vodnika:


```
ListPlot [Table[{100 yC[[i]],  $\sigma[[i]] * 10^9$ }, {i, 2 nx + ny + 1, N0}], PlotRange -> {-3.7, 0},
Frame -> True, Axes -> None, Times14, FrameLabel -> {"y / cm", " $\sigma / (nC/m^2)$ ", "", ""}]
```



Porazdelitev elektrine na celotnem obodu vodnika (v odvisnosti od dolzine poti od levega spodnjega vogala po obodu v matematicni pozitivni smeri):

```
PorazdElektr = ListPlot [
Table [{100 If [i == 1,  $\delta l [[1]] / 2$ ,  $\sum_{j=1}^{i-1} \delta l [[j]] + \delta l [[i]] / 2$ ],  $\sigma[[i]] * 10^9$ }, {i, 1, N0}],
Frame -> True, Axes -> None, Times14, FrameLabel -> {"l / cm", " $\sigma / (nC/m^2)$ ", "", ""}];
```

```
Show[PorazdElektr, Graphics[{{Text["spodaj", {0.5, 2.9}],
Text["desno", {2, 2.9}], Text["zgoraj", {3.5, 2.9}], Text["levo", {5, 2.9}]}]]]
```



■ Izracun potenciala in risanje ekvipotencialk

Dolzina distancnega vektorja med središčno točko i -tega segmenta in splosno točko (x, y) , v kateri bomo racunali potencial:

$$\text{absP}[i_, \mathbf{x}_-, \mathbf{y}_-] := \sqrt{(\mathbf{x} - \mathbf{xC}[[i]])^2 + (\mathbf{y} - \mathbf{yC}[[i]])^2}$$

Potencial splosne točke (x, y) :

$$V[\mathbf{x}_-, \mathbf{y}_-] := \frac{-1}{2\pi * \epsilon_0} * \sum_{i=1}^{NO} \sigma[[i]] * \delta l[[i]] * \text{Log}[\text{absP}[i, \mathbf{x}, \mathbf{y}]] - E_0 * \mathbf{x}$$

Potencial je tu izracunan le do konstante (C_0) natančno (torej kot ce bi bila $C_0 = 0$).

Pa se to: zgorja formula za potencial je singularna v središnih tockah segmentov. Razmislite, kako bi resili to tezavo?

Presek vodnika, ki ga bomo vrisali v sliko ekvipotencialk:

```
vodnik = Line[100 {{0, 0}, {a, 0}, {a, b}, {0, b}, {0, 0}}];
```


Obmocje risanja definirajmo s parametri x_{\min} , x_{\max} , y_{\min} in y_{\max} :

```
xmin = -0.04;
xmax = 0.05;
ymin = -0.04;
ymax = 0.06;
```

Ekvipotencialke elektricnega potenciala v okolici vodnika:

```
ekvipot = ContourPlot[V[x/100, y/100], {x, 100 xmin, 100 xmax}, {y, 100 ymin, 100 ymax},
PlotPoints -> 100, Contours -> 43, ContourShading -> False, AspectRatio -> (ymax - ymin) / (xmax - xmin),
PlotRange -> 100 {{xmin, xmax}, {ymin, ymax}}, ContourStyle -> AbsoluteThickness[0.2]];
```

```
Show[ {ekvipot, Graphics[ {AbsoluteThickness[1.5], vodnik} ]},
Frame -> True, Times14, FrameLabel -> {"X / CM", "y / CM", "", ""} ]
```



Izračunajmo se potencial samega vodnika (oz. katerekoli točke v območju vodnika) ter primerjajmo relativna odstopanja (v procentih) med izračunanimi vrednostmi v različnih točkah znotraj vodnika:

```
Vvod = V[a / 2, b / 2]
```

```
- 0.5
```

```
{V[a / 4, b / 4], (V[a / 4, b / 4] - Vvod) * 100 / Vvod}
```

```
{-0.499684, -0.0632821}
```

```
{V[3 * a / 4, b / 2], (V[3 * a / 4, b / 2] - Vvod) * 100 / Vvod}
```

```
{-0.500128, 0.0255658}
```

```
{V[0.001 * a, b / 2], (V[0.001 * a, b / 2] - Vvod) * 100 / Vvod}
```

```
{-0.498976, -0.204724}
```

```
{V[0.999 * a, 0.999 * b], (V[0.999 * a, 0.999 * b] - Vvod) * 100 / Vvod}
```

```
{-0.509938, 1.98757}
```

Opazimo lahko, da je izračunana vrednost potenciala povsod znotraj vodnika približno enaka. Se največja napaka je v točkah blizu površine vodnika (torej blizu enojnega sloja elektronov). Opazimo lahko se, da vodnik zavzame takšen potencial, kot ga je pred vstavitvijo vodnika (v prvotno homogeno polje) imela (bodoca) simetralna ravnina vodnika $x = a/2$:

```
V0 = -E0 * (a / 2)
```

```
- 0.5
```

Tudi tukaj je potencial izračunan le do konstante (C_0) natančno (torej kot ce bi bila $C_0 = 0$).

Shranjevanje resitve

Preverimo, v kateri direktorij bo shranjena resitev (z ukazom `SetDirectory` ga lahko spremenimo):

```
Directory []
```

Shranimo celoten izracun:

```
DumpSave ["VodnikPravokPresekVHomogPolju.mx", "Global`"];
```

Ko izracuni trajajo daljsi cas, nam pride zelo prav, da resitev shranimo in jo lahko pozneje preberemo in uporabimo, ne da bi ponovno zaganjali izracune. Preberemo jo takole:

```
<< VodnikPravokPresekVHomogPolju.mx
```